Appended Numbers Game
APL (66)

{↑¯5↑{(⍕⍵),': ',⍺}/∆,⍪⍳⍴∆←⍺{(1&lt;⍴⍵)∧⍺&gt;0:∆,(⍺-1)∇⊃∆←,/⍕¨2+/⍎¨⍵⋄⍬}⍕⍵}


The left argument is the maximum iteration count and the right argument is the start number.

Explanation:


∆←⍺{...}⍕⍵: pass the left argument as a number and the right argument as a string to the function that calculates the list of numbers, and store it in ∆:

(1&lt;⍴⍵)∧⍺&gt;0:: if the amount of digits is more than 1 and the amount of iterations left is more than 0:

⍎¨⍵: evaluate each digit
2+/: sum each pair
⍕¨: format each number as a string
∆←,/: concatenate the strings and store in ∆
∆,(⍺-1)∇⊃∆: return ∆, followed by the result of applying this function to ∆ with one less iteration allowed

⋄⍬: if not, return the empty list

∆,⍪⍳⍴∆: pair each element of ∆ with its index in ∆ 
{...}/: for each pair:

(⍕⍵),': ',⍺: return a string with the index, followed by :, followed by the number

↑¯5↑: turn the list of strings into a matrix so they display on separate lines, and take the last 5 items


Test:

      5{↑¯5↑{(⍕⍵),': ',⍺}/∆,⍪⍳⍴∆←⍺{(1&lt;⍴⍵)∧⍺&gt;0:∆,(⍺-1)∇⊃∆←,/⍕¨2+/⍎¨⍵⋄⍬}⍕⍵}3541
1: 895 
2: 1714
3: 885 
4: 1613
5: 774 
      50{↑¯5↑{(⍕⍵),': ',⍺}/∆,⍪⍳⍴∆←⍺{(1&lt;⍴⍵)∧⍺&gt;0:∆,(⍺-1)∇⊃∆←,/⍕¨2+/⍎¨⍵⋄⍬}⍕⍵}3541
6: 1411
7: 552 
8: 107 
9: 17  
10: 8